

Apéndice 5

Uso de verilog

A5.1 Introducción

Es uno de los lenguajes de descripción de hardware (HDL) más utilizados.

Las herramientas CAD soportan el ingreso de diseños digitales en base a esquemáticos, los cuales se arman conexionando componentes básicas o de bibliotecas en un ambiente visual.

Sin embargo a medida que el sistema aumenta su tamaño, cada vez es más difícil la descripción por esquemáticos. Los esquemáticos describen la conectividad y por lo tanto el armado del sistema, pero no destacan o enfatizan la funcionalidad de éste.

Los lenguajes descriptivos de hardware permiten diseñar, en forma abstracta, complejos sistemas digitales que luego de ser simulados podrán ser implementados en dispositivos programables como FPGA o CPLD, lo cual se logra describiendo el sistema digital mediante código HDL.

Las herramientas computacionales de ayuda al diseño (CAD) permiten diseñar, verificar códigos de alto nivel, simular el circuito representado y crear los códigos binarios que serán cargados en los dispositivos programables (síntesis).

Debido a que los módulos pueden reutilizarse, y los subsistemas más frecuentemente utilizados pueden estar prediseñados, la programación a través de lenguajes toma cada vez más importancia en el diseño de sistemas digitales.

Los lenguajes HDL deben permitir la **simulación**, cuyos objetivos son: verificar que el diseño es funcionalmente correcto; es decir que se cumplan las **especificaciones lógicas**; y además que se cumplan las **especificaciones temporales**; es decir que los tiempos de propagación a través de las componentes y de los caminos que permiten la interconexión cumplan las especificaciones de setup y hold de los registros, en caso de sistemas secuenciales sincrónicos. Como deben considerarse todos los casos, es difícil asegurar que la simulación entregará la solución definitiva, ya que el conjunto de estímulos, para los cuales se realiza la simulación, puede no ser completo. Al mismo tiempo los lenguajes deben realizar la **síntesis lógica**; es decir traducir el programa a una serie de compuertas y flip-flops; también deben efectuar minimizaciones, y permitir **mapear a los bloques tecnológicos disponibles**; para finalmente **colocar y enrutar** de forma conveniente los bloques en el dispositivo programable.

Verilog existe desde el año 1984, luego de varios años de uso se ha estandarizado y su última versión es del año 2001. Es un lenguaje similar a C, y dentro de los HDL, además de las

descripciones abstractas permite representaciones en bajo nivel; es decir puede describir sistemas digitales en base a compuertas, e incluso en base a transistores.

Verilog permite que en un diseño se puedan usar diferentes niveles de descripción de sistemas digitales en un mismo ambiente; las diferentes descripciones pueden ser simuladas para verificar el funcionamiento y además pueden ser sintetizadas; es decir traducidas a la interconexión de componentes básicas de un dispositivo programable.

Verilog permite la descripción **estructural** del diseño en base a componentes básicas, y descripciones más abstractas que se enfocan en la **conducta** del sistema. La conducta puede describirse mediante expresiones lógicas y también empleando procedimientos.

Un diseño basado en descripciones funcionales o de comportamiento puede resultar lento y de gran costo en área. Las descripciones en niveles estructurales permiten optimizar los circuitos lógicos para maximizar la velocidad y minimizar el área.

Se expondrán las formas de descripción para sistemas combinacionales y secuenciales. La Figura A5.1 resume las principales etapas del diseño digital.




Figura A5.1

A5.2 Descripción estructural.

A5.2.1. Nivel compuertas.

Permite representar una red lógica mediante sus ecuaciones. Para esto se emplean **funciones lógicas** básicas para representar compuertas.

Los operadores and, or not, en minúsculas, son palabras reservadas, y se emplean como nombres de funciones.

La función and [nombre de compuerta] (salida, entrada1, entrada2); describe un and de dos entradas. Donde el nombre de compuerta puede ser opcional. Importante es destacar que estas compuertas pueden tener “muchas” entradas.

La función or(x, ~a, b, ~c); representa a la compuerta or de tres entradas que puede describirse por la ecuación lógica: $x = \bar{a} + b + \bar{c}$

Las compuertas definidas son: and, nand, or, nor, xor, xnor, buf, not. La salida es el primer argumento de la lista.

El inversor y el buffer pueden tener varias salidas, el último elemento de la lista es la entrada.

Para representar un sistema se requiere definir un **módulo**, el que se describe como una función del lenguaje C; en la cual los argumentos son los nombres de las variables de entrada, de salida y bidireccionales del módulo.

Luego deben colocarse **listas de variables** de entrada, de salida y bidireccionales. El módulo debe estar terminado con la sentencia **endmodule**.

Pueden emplearse comentarios, para mejorar la legibilidad.

Los identificadores de variable deben comenzar con una letra o el carácter ‘_’. Cuando se definen identificadores se consideran distintas las letras minúsculas y mayúsculas.

La Figura A5.2 representa una red lógica con dos salidas y cuatro entradas. Su descripción puede efectuarse mediante un módulo.




Figura A5.2

```

module bloque(a, b, c, d, f1, f2); //note que termina en punto y coma.
  input a, b, c, d; // comentario de fin de línea
  output f1, f2;
/*

```

```

  Aquí se describen las funciones lógicas mediante compuertas o expresiones.
*/

```

```

endmodule

```

Ejemplo A5.1.

Se tiene la función de cuatro variables:

$$Z(A, B, C, D) = \sum m(0, 3, 5, 12, 13) + \sum d(1, 2, 15)$$

Empleando alguna herramienta de minimización (ver A3.2 en Apéndice 3) se obtiene la forma minimizada en suma de productos:

$$Z = B\bar{C}D + \bar{A}\bar{B} + AB\bar{C}$$

La ecuación se puede traducir en el módulo siguiente, notar el empleo de los nodos internos: n1, n2 y n3 para formar la salida Z.

```
module ejemploA5_1(A, B, C, D, Z);
input A, B, C, D;
output Z;
  and(n1, B, ~C, D);
  and(n2, ~A, ~B);
  and(n3, A, B, ~C);
  or(Z, n1, n2, n3);
endmodule
```

Con fines de simulación puede asociarse un **retardo de propagación** a cada compuerta. Por ejemplo, la descripción: and #10 (x, y, z) indica que esa compuerta and tiene asociado un retardo de 10 unidades de tiempo de simulación. También puede darse un nombre a cada compuerta, escribiéndolo antes de la lista de puertas.

En un proceso de síntesis no es recomendable asociar retardos a las compuertas, ya que los algoritmos pueden aumentar su tiempo de ejecución buscando satisfacer los requerimientos de tiempo de las compuertas.

Ejemplo A5.2.

Mux de dos vías a una, implementado mediante compuertas.




Figura A5.3

La ecuación del mux de la Figura A5.3, como suma de productos, puede expresarse según:

$$out(c0, x1, x0) = c0 x1 + c0' x0$$

El módulo puede escribirse directamente a partir de la ecuación.

```
// Multiplexer nivel compuertas. multiplexer 2-1, vías de 1-bit
// Cuando c0=1, out=x1; y cuando c0=0, out=x0
```

```
module mux21_gt (out, c0, x1, x0);
    output out;          // mux output
    input  c0, x1, x0;    // mux inputs
    wire n1, n0;          // nodos internos
    and(n0, x0, ~c0);
    and(n1, x1, c0);
    or(out, n1, n0);

endmodule
```

Las conexiones o **nets** permiten ir uniendo componentes de un diseño digital. En el ejemplo anterior; n1 y n0, los nodos internos, son declarados **wire** (en español alambres) y son un tipo de net. Las señales tipificadas con wire, no tienen capacidad de almacenamiento; se consideran continuamente alimentadas por la salida de un módulo o por una asignación continua assign. Si las señales de tipo wire de entrada se dejan desconectadas, toman el valor x, que significa desconocido. Una variable de tipo wire es simplemente un rótulo sobre un alambre.

Una señal con **tipo de datos** de modo wire representa la actualización continua en circuitos combinacionales de las salidas, respecto a cambios de las entradas. Otro tipo de datos, que suele estar asociado a salidas de flip-flops es **reg**, que es abreviatura de registro.

A5.2.2. Nivel transistores.

Verilog permite descripciones a nivel de conmutación de transistores, lo cual permite modelar compuertas.

Ejemplo A5.3.

Descripción de inversor CMOS en nivel de conmutación.




Figura A5.4

La Figura A5.4, muestra las señales empleadas en el módulo. Los tipos de conexiones o **nets**: supply1 y supply0 conectan señales a un voltaje de polarización y a tierra, respectivamente.

Los transistores cmos se especifican con tres argumentos (salida, entrada, control), y se dispone de las formas nmos y pmos.

El siguiente módulo implementa al inversor cmos de la Figura A5.4.

```
module inv_sw (out, in);
    output out;          // salida inversor
    input in;            // entrada inversor

    supply1 power;        // "power" conexión a Vdd . Tipos de net.
    supply0 ground;       // "ground" conexión a Gnd

    pmos (out, ground, in); // instancia switch pmos
    nmos (out, power, in);  // instancia switch nmos

endmodule
```

Ejemplo A5.4.

Multiplexor basado en compuertas de transmisión.

La Figura A5.5 muestra a la izquierda un símbolo para la compuerta de transmisión y, a la derecha, su implementación basada en un par de transistores. Cuando ctrl=0 la salida está en alta impedancia; cuando ctrl=1 la salida es igual a la entrada.




Figura A5.5

Verilog dispone de un modelo intrínseco para la compuerta de transmisión. Para instanciar una compuerta de transmisión se emplea:

```
cmos [nombreinstancia]([output],[input],[nmosgate],[pmosgate])
```

La Figura A5.6 muestra un multiplexor de dos vías a una. Cuando ctrl=1, se tiene que out=in2; y cuando ctrl=0 se tendrá out=in1.




Figura A5.6

La implementación de este multiplexor se logra con el siguiente módulo, en el cual se ha generado la señal *ctrl'* a partir de *ctrl* y un inversor; el cual se instancia según el modelo anterior y con nombre *I1*. Se definen dos compuertas de transmisión con nombres *C1* y *C2*.

```
// Multiplexer nivel conmutación. 1-bit 2-1 multiplexer
// Cuando ctrl=0, out=in1; y cuando ctrl=1, out=in2

module mux21_sw (out, ctrl, in1, in2);

    output out;           // mux output
    input  ctrl, in1, in2; // mux inputs
    wire    w;           // nodo interno de tipo wire

    inv_sw I1 (w, ctrl);  // instancia inversor

    cmos C1 (out, in1, w, ctrl); // instancias switches cmos
    cmos C2 (out, in2, ctrl, w);

endmodule
```

A5.3 Descripción del comportamiento (Behavior).

Cuando son numerosas las ecuaciones de la red puede ser muy laboriosa la descripción de la estructura mediante compuertas.

Los lenguajes permiten una descripción más abstracta, y a la vez compacta, de las redes booleanas, ya que puede describirse el comportamiento o la conducta de la red.

Se describe lo que debe efectuar el sistema, empleando sentencias del lenguaje; es decir, la red booleana se describe como un programa.

Lo que el programa describe en este nivel son los registros y las transferencias y transformaciones de vectores de información entre los registros; este nivel se denomina **RTL** (Register Transfer Level).

También podría decirse que la descripción del comportamiento es una descripción de la **arquitectura** del sistema digital.

El diseñador se puede concentrar en el análisis de arquitecturas alternativas, mediante la simulación, determinando que partes del diseño requerirán un estudio más detallado.

Este lenguaje permite **describir con precisión la funcionalidad** de cualquier diseño digital.

La descripción del comportamiento puede efectuarse de dos modos: Mediante **expresiones** lógicas o mediante **procedimientos**.

El compilador del lenguaje traduce estas representaciones abstractas a ecuaciones lógicas y éstas a su vez son mapeadas a los bloques lógicos del dispositivo programable.

Ejemplo A5.5. Expresiones.

Mediante la **asignación continua**, y escribiendo expresiones lógicas, empleando los operadores lógicos al bit del lenguaje C, puede obtenerse la misma funcionalidad de la descripción de Z mediante compuertas, vista en el Ejemplo A5.1.

```
module ejemploA5_5(input A, B, C, D, output Z);

    assign Z = (B & ~C & D) | (~A & ~B) | (A & B & ~C);

endmodule
```

Ejemplo A5.6. Mux mediante Expresiones.

El mux visto antes, en el Ejemplo A5.2, puede describirse:

```
// Multiplexer nivel expresiones . 1-bit 2-1 multiplexer
// Cuando c0=1, out=x1; y cuando c0=0, out=x0
module mux21_exp (output out, input c0, x1, x0);

    assign out= (c0&x1) | (~c0&x0);

endmodule
```

Operadores.

Los **operadores al bit** del lenguaje C: ~ para el complemento a uno, & para el and, | para el or, y ^ para el xor, efectúan la operación bit a bit del o los operandos. Además se dispone de los siguientes: ~& para el nand, ~| para el or, y: ~^ o ^~ para la equivalencia o not xor.

Se dispone de operadores unarios que efectúan la **reducción del operando a un bit**. Lo que efectúan es aplicar el operador a todos los bits de la señal. Se tienen: & para la reducción AND, | para la reducción OR, ^ para la reducción XOR, ~& para la reducción NAND, ~| para la reducción NOR, y ~^ para la reducción XNOR.

Verilog dispone de dos operadores adicionales, no incluidos en el lenguaje C, que son la igualdad y desigualdad, considerando los valores no conocido x, y alta impedancia z. En A5.6 se describe la forma de escribir valores. Se anotan: (a == b) y (a != b), en estas operaciones si se incluyen bits con valores x o z, también deben los correspondientes bits ser iguales para que la comparación sea verdadera, valor 1; o valor 0 en caso contrario.

En los test de igualdad o desigualdad (a == b) y (a != b), que están presentes en C, si existen operandos con valores x o z, el resultado es desconocido, es decir x.

Ejemplo A5.7. Procesos.

La descripción de la conducta mediante **procedimientos** está basada en la definición de un bloque **always**. Esta construcción está acompañada de una **lista de sensibilidad**, cuyo propósito es evaluar las acciones dentro del procedimiento **siempre que** una o más de las señales de la lista cambie.

Si se produce un evento de la lista de sensibilidad y se desea que la salida cambie inmediatamente, se efectúa una asignación **bloqueante** (se usa el signo igual); esto modela la conducta de un proceso combinacional. El calificativo bloqueante deriva del modelo de simulación orientado a eventos del lenguaje Verilog; en este sentido se dice bloqueante, ya que todos los otros eventos quedan pendientes hasta que se produzca la renovación del valor; es decir, esta asignación no puede ser **interrumpida** por otras órdenes Verilog concurrentes.

Si se produce un evento de la lista de sensibilidad y se desea que la salida cambie luego de un tiempo, se efectúa una asignación **nobloqueante** (se usa el signo <=), esto modela el proceso de escritura en un flip-flop. Apenas producido el evento, se recalculan los lados derechos de las asignaciones nobloqueantes; una vez que se han actualizado todos los valores, correspondientes a ese paso de simulación, se copian en las variables a la izquierda.

Importa el orden en que se efectúan una serie de asignaciones bloqueantes; ya que las acciones dentro de un bloque de un proceso se ejecutan secuencialmente. Cuando se tiene una serie de asignaciones nobloqueantes, no importa el orden en que son escritas.

Cada bloque always define un **hebra de control**. Si hay varios bloques always en un módulo, se considera que están corriendo **concurrentemente**.

Las acciones dentro del bloque o proceso son evaluadas en orden secuencial. En esta construcción puede tenerse uno o más bloques always dentro del módulo; y por lo menos una acción en uno de los bloques always. Si dentro del procedimiento se asigna valores a una de las señales, ésta debe ser declarada de tipo **reg**.

```
module ejemploA5_7(A, B, C, D, Z);
```

```

input A, B, C, D;
output Z;
  reg Z;    //se marca como registrada pero es combinacional.

  always @(A or B or C or D)
  begin
    Z = (B & ~C & D) | (~A & ~B) | (A & B & ~C);
  end
endmodule

```

Cuando se efectúa la síntesis de esta descripción, Z será la salida de compuertas, y no se usará un flip-flop para guardar el valor de Z.

Las entradas deben estar en la lista de sensibilidad, las salidas deben ser asignadas en cada una de las alternativas controladas por sentencias.

En forma similar se dispone de **initial** que crea un proceso que se realiza sólo una vez, lo cual permite inicializar variables. El proceso always se ejecuta **siempre**.

La diferencia fundamental entre el tipo net y reg es que a este último se le asignan valores en forma explícita. El valor se mantiene hasta que se efectúe una nueva asignación. Las salidas de flip-flops serán de tipo reg.

Ejemplo A5.8. Mux mediante Procesos.

El mux visto en los Ejemplos A5.2 y A5.6, puede describirse mediante un proceso, según:

```

// Multiplexer nivel proceso . 1-bit 2-1 multiplexer
// Cuando c0=1, out=x1; y cuando c0=0, out=x0

module mux21_proc (out, c0, x1, x0);
  output out;    // mux output
  input  c0, x1, x0; // mux inputs
  reg out;
  always @(*) //asterisco representa todas las entradas que pueden modificar la salida
    out= (c0&x1) | (~c0&x0);
endmodule

```

Ejemplo A5.9. Mux mediante lenguaje.

La introducción de elementos similares a los empleados en lenguajes de programación permite describir procesos en forma cada vez más abstracta. Como veremos esto facilita la generalización.

```

// Multiplexer nivel proceso abstracto. 1-bit 2-1 multiplexer
// Cuando c0=1, out=x1; y cuando c0=0, out=x0

module mux21_procAbstracto (out, c0, x1, x0);

```

```

output out;          // mux output
input  c0, x1, x0;    // mux inputs
reg out;
always @(*)
    if (c0==1) out=x1; else out=x0;
endmodule

```

En descripciones del comportamiento, mediante lenguaje, se representa la funcionalidad del módulo, independiente de la implementación lógica en base a compuertas. La herramienta CAD debería resolver el descenso o **compilación a compuertas**.

Sentencias.

Los modelos que describen el comportamiento contienen procesos que controlan la simulación y que manipulan variables. Cada proceso tiene su propio flujo de acciones. Las acciones de cada proceso se ejecutan concurrentemente, permitiendo modelar sistemas digitales que son inherentemente concurrentes o paralelos.

Se dispone de similares construcciones de **secuenciación de acciones** que en el lenguaje C. Algunos detalles son diferentes: si se emplea una variable índice entera en un for debe ser declarada de tipo integer (no int); no se emplea ++ para incrementar variables, sino var=var+1; se emplea begin .. end en lugar de {..}.

En la sentencia case no son necesarios los break; en los casos pueden colocarse valores con x o z, y sólo si sintonizan exactamente, se realiza la acción asociada al caso.

Existen adicionalmente casez y casez que permiten comparación con condiciones superfluas con valores que contengan x o z.

La ejecución de una acción puede ser retardada hasta que una condición se haga verdadera. Esto se logra con la construcción: **wait** (condición) acción;

Se dice que la sentencia wait es sensible a los **cambios de nivel**, a diferencia de los eventos controlados por @, que disparan la acción con los **cantos** de las señales de la lista.

Se evalúa la condición del wait, y si es falsa la acción se detiene; es decir se espera hasta que la condición sea verdadera, momento en el cual se realiza la acción.

Ejemplo de uso:

Cuando se habilita la escritura, con WEnable=1, se transfiere el contenido del busA al registro regA, puede escribirse:

```
wait (!WEnable) regA <= busA;
```

Sincronización entre procesos.

A los eventos por cantos puede dárseles un nombre.

Por ejemplo la transferencia: @ trigger rega <= regb;

Puede ser realizada desde otro proceso, mediante: -> trigger;

Acciones en paralelo.

Entre las palabras `fork` y `join` las acciones se consideran de ejecución paralela. Los separadores `begin`, `end` encierran sentencias secuenciales. Si las sentencias tienen retardos, en un conjunto de acciones en paralelo, éstos se consideran respecto del inicio del bloque.

A5.4 Diseños jerárquicos.

Verilog permite descripciones **jerárquicas** de redes complejas. Por ejemplo un sumador completo de 4 bits puede describirse mediante la instanciación de cuatro módulos con las variables actuales de un sumador completo de un bit.

Si se tiene en un archivo, denominado `sumadorcompleto.v`, el siguiente módulo combinacional definido por asignaciones continuas:

```
module sumadorcompleto(cin, a, b, sum, cout);
    input cin, a, b;
    output sum, cout;

    assign sum = a ^ b ^ cin,      // Operador or exclusivo
    assign cout = (a & b) | (a & cin) | (b & cin);
endmodule
```

El cual puede representarse en el esquema de la Figura A5.7.




Figura A5.7

Alternativamente podría haberse efectuado una descripción estructural:

```
module sumadorcompleto (Cin, a, b, sum, Cout);
    input Cin, a, b;
    output sum, Cout;
    wire n1, n2, n3;
    xor (sum, a, b, Cin);
    and (n1, a, b);
    and (n2, a, Cin);
    and (n3, b, Cin);
    or (Cout, n1, n2, n3);
endmodule
```

El sumador de 4 bits, puede estar en otro archivo, en el cual se incluye la definición del módulo básico. Se le da un nombre a cada instancia, etapa_i en el ejemplo, el que debe estar precedido por el nombre del módulo básico.

En la Figura A5.8, se ilustran las instancias y las correspondencias de los nombres de las puertas de cada etapa con los nombres del módulo básico. Con este diagrama resulta sencillo efectuar las instancias dentro del módulo.




Figura A5.8

```
`include "sumadorcompleto.v"
```

```
module fulladder4 (Cin, a3, a2, a1, a0, b3, b2, b1, b0, sum3, sum2, sum1, sum0, Cout);
```

```
  input Cin, a3, a2, a1, a0, b3, b2, b1, b0;
```

```
  output sum3, sum2, sum1, sum0, Cout;
```

```
  sumadorcompleto etapa0 (Cin, a0, b0, sum0, c1);
```

```
  sumadorcompleto etapa1 ( c1, a1, b1, sum1, c2);
```

```
  sumadorcompleto etapa2 ( c2, a2, b2, sum2, c3);
```

```
  sumadorcompleto etapa3 ( c3, a3, b3, sum3, Cout);
```

```
endmodule
```

Los módulos no deben ser considerados procedimientos o funciones, en el sentido que no son invocados. Se dice que son instanciados al inicio del programa y permanecen hasta el final del programa. En el ejemplo anterior, fulladder4 crea 4 sumadorescompletos, denominados etapa0 hasta etapa3.

A5.5. Fundamentos de simulación.

El lenguaje debe proveer facilidades para generar estímulos, medir respuestas, y efectuar despliegues de los datos.

Empleando programación jerárquica un esquema que mantiene separado el módulo del diseño con el módulo de test, se visualiza en la Figura A5.9. La ventaja de esto es que el módulo con el diseño debe luego sintetizarse; de esta forma todas las sentencias que tienen que ver con la simulación quedan definidas dentro del módulo test; la finalidad de éste es generar los estímulos y grabar los resultados de los test. El módulo test podría describir un elemento de hardware, por ejemplo si se generaran los estímulos posibles mediante un contador o una memoria.

La activación del diseño y test se logra con un tercer módulo denominado Banco de prueba ubicado en el tope de la jerarquía, en el diagrama de la Figura A5.9.




Figura A5.9.

Ejemplo A5.10. Banco de prueba para semisumador binario.

El módulo del semisumador escrito en forma estructural:

```
module semisumador(a, b, suma, cout);
input a, b;
output suma, cout;
  xor(suma, a, b); and(cout, a, b);
endmodule
```

En la confección del test del semisumador se define un bloque que sólo se efectuará una vez, esto se logra con la palabra reservada **initial**. El símbolo: #10 b=1; indica efectuar un **retardo** de 10 (unidades de tiempo de simulación) y luego efectuar la asignación. Con la secuencia dada se generan los estímulos para las entradas del semisumador, con valores para ab: 00, 01, 11, 10. La palabra **finish** termina la generación de estímulos.

La palabra \$time muestra el tiempo en unidades de simulación.

El comando **monitor** muestra en la salida estándar, en formato similar a printf, los valores de la lista de variables; las comas adicionales se interpretan como espacios.

Cuando cambia uno de los valores de la lista del comando monitor, se produce impresión de una línea. Los valores se imprimen cuando todos los eventos del paso se han producido; es decir se imprime al final de un intervalo de simulación. El formato %b despliega en binario, %d en decimal, %h en hexadecimal. Sólo un monitor puede estar activo.

```
module testsemisumador(suma, cout, a, b);
input suma, cout;
output a, b; //genera estímulos de entrada para el diseño
reg a, b;
```

```

initial
begin
  $monitor("t=", $time, "a=%b, b=%b, suma=%b, cout=%b", a, b, suma, cout);
  a=0; b=0;
  #10 b=1;
  #10 a=1;
  #10 b=0;
  #10 $finish;
end
endmodule

```

En Verilog se forma bloques de instrucciones, encerrándolas entre las palabras reservadas: `begin`, `end`; esto es una diferencia con el lenguaje C que usa paréntesis cursivos.

El módulo `BancoDePrueba`, en el tope de la jerarquía, define las señales que se emplearán, las declara de tipo **wire**.

```

module BancoDePrueba;
wire a, b, su, co;
  semisumador    sm(a, b, su, co);
  testsemisumador ts(su, co, a, b);
endmodule

```

Efectuando la simulación, se obtiene el listado:

```

t=      0 a=0, b=0, suma=0, cout=0
t=     10 a=0, b=1, suma=1, cout=0
t=     20 a=1, b=1, suma=0, cout=1
t=     30 a=1, b=0, suma=1, cout=0

```

Listado que permite verificar, en forma exhaustiva, el funcionamiento del semisumador. La prueba es completa, en este ejemplo, ya que se generan todas las combinaciones de estímulos posibles; en este caso se está verificando la tabla de verdad.

Funcionamiento del simulador.

En forma simplificada el funcionamiento del simulador es como sigue:

Initial y always comienzan a ejecutarse en el tiempo $t=0$, y continúan hasta encontrar un retardo (`#`); cuando esto ocurre, la condición que lo produce se almacena en la cola de los que esperan, por el tiempo indicado; luego del cual vuelve a colocarse el bloque en la cola de eventos activos para atender; es decir reasume la ejecución del bloque desde donde fue detenido.

Los siguientes **eventos activos** se itineran en cualquier orden: Salidas de compuertas, asignaciones continuas y bloqueantes, lectura de entradas y actualización de salidas, ejecución de comandos display, cálculo de lados derechos de asignaciones nobloqueantes.

Cuando, para un intervalo de tiempo de simulación, no quedan eventos activos por ser atendidos, se procesa la **cola de eventos de asignaciones nobloqueantes**, pasando al lado izquierdo los valores ya calculados.

Luego de lo cual se procesan los **eventos de monitor**, los comandos `strobe` y `monitor`. Después de esto se pasa al procesar los eventos asociados al siguiente intervalo de tiempo.

Los comandos `$monitor`, `$display` y `$strobe` operan en forma similar a `printf`.

El comando **\$monitor** muestra los valores actualizados de todas las variables de su lista, al final de un paso de simulación y después que todas las asignaciones para ese paso de simulación hayan sido efectuadas.

El comando **\$display** muestra los cambios que se producen antes de que las asignaciones nobloqueantes sean actualizados, ya que se considera evento activo.

El comando **\$strobe** permite mostrar las asignaciones nobloqueantes una vez realizadas.

Cuando un proceso está en ejecución sólo se detiene al encontrar un retardo (`#`) o un evento de una lista de sensibilidad (`@`) o el comando `wait`, este último comando espera hasta que su variable argumento sea verdadera.

Un bloque `always` o proceso debe contener alguno de los elementos de control de tiempo para que la simulación se detenga. El bloque continúa su ejecución: cuando transcurre el retardo, o cuando cambia una de las señales de la lista de sensibilidad o cuando ocurre el evento que espera `wait`. Un evento produce otros eventos. Son éstas las únicas instancias en que se asume que el tiempo transcurre; el resto de los comandos (`if-else`, `case`, etc.) se considera que son instantáneos.

Los procesos detienen la simulación en determinados instantes, y también lo hacen las activaciones de salidas de compuertas.

Los eventos son itinerados para tiempos específicos y son colocados en una cola de espera, ordenados en el tiempo, para ser atendidos. Los primeros eventos quedan al frente, los últimos al fondo. Estando en el intervalo presente pueden procesarse varios eventos, no importando el orden; una vez atendidos todos los eventos asociados al intervalo presente, se pasa al siguiente. Durante el procesamiento pueden generarse nuevos eventos, que son ubicados en los lugares temporales que les corresponden para su posterior procesamiento.

Verilog emplea la sintaxis de C, pero es un lenguaje que permite describir **procesos concurrentes**, los eventos descritos en el párrafo anterior no están implementados en C, que es un lenguaje procedural.

La simulación en Verilog se produce por eventos que ocurren en determinados instantes; es una **simulación por eventos discretos**.

Ejemplo A5.11. Simulación nivel transistores.

```
// Estímulos para verificar el inversor CMOS nivel de transistores del Ejemplo A5.3.
module stimulus;
    reg          in;          // se escribe en in. Se le da tipo reg.
```



```

wire      out;

inv_sw I0 (out, in); // instancia inversor

initial begin
    in=0; $display("%b, %b, t=%0d", in, out, $time);
    #1 in=1'b1; $display("%b, %b, t=%0d", in, out, $time);
    #4 in=1'b0; $display("%b, %b, t=%0d", in, out, $time);
    #5 in=1'b1; $display("%b, %b, t=%0d", in, out, $time);
    #5 in=1'b0;
    #5 $finish;
end
endmodule

```

Generando la siguiente traza:

```

0, x, t=0    //Al inicio, en t=0, aún no hay salida. No se conoce es x.
1, 0, t=1
0, 1, t=5
1, 0, t=10

```

El descriptor 0 antes de d, en el string de control del display, especifica no generar espacios, e imprimir el tiempo en ancho variable, de acuerdo a la magnitud.

El siguiente proceso da un máximo tiempo a la simulación.

```

initial begin: DetenerSimulacion
// Detiene la simulación después de 400 unidades de tiempo de simulación.
#400; $stop;
end

```

A5.6. Buses, registros, vectores, arreglos. Valores constantes.

En numerosas situaciones de sistemas digitales, las variables de entrada y salida son multivaluadas, sin embargo ha sido tradicional codificarlas como **vectores booleanos**.

En general cuando se agrupan una serie de señales con un nombre común se origina un **bus** o vector booleano.

En Verilog, los **tipos de datos** net y reg pueden ser declarados como vectores. Existen mecanismos de acceso a partes o a la totalidad del vector

Puede convenirse en establecer el bit más significativo del vector como el que se escribe primero; además importa el orden con el que se declara el ancho del bus.

El siguiente ejemplo define out como la salida de un registro de 8 bits.

```
reg [7:0] out;
```

Se tiene que el bit 7 es el más significativo, el 0 el menos significativo. Convenio little endian.

Valores.

Para describir **constantes** o valores se emplea la notación: <largo>'<base><valor>. Donde la base suele ser binaria y hexadecimal; si no se coloca '<base>' se asume decimal. A los valores binarios 1 y 0, se agregan los valores **z** y **x** que representan **alta impedancia** y **desconocido**, respectivamente.

Con estos nuevos valores, la operación que realiza una compuerta and se rige por la siguiente tabla de verdad.

and	0	1	x	z
0	0	0	0	0
1	0	1	x	x
x	0	x	x	x
z	0	x	x	x

Figura A5.10.

Para una compuerta or se tiene:

or	0	1	x	z
0	0	1	x	x
1	1	1	1	1
x	x	1	x	x
z	x	1	x	x

Figura A5.11.

Los bits de un vector o bus pueden ser referenciados empleando la notación: [<bit de partida>:<bit de término>]

De esta forma puede inicializarse un registro en forma parcial, asignado un valor a los primeros cuatro bits de la variable out, mediante:

```
out[7: 4] = 4'b1010;
```

La operación **concatenación**, agrupa los elementos separados por comas en una lista definida entre paréntesis cursivos.

```
out = {A[3,0], 4b'0101};
```

Adicionalmente puede emplearse el comando de **repetición** que especifica el número de veces que se repite una concatenación: {2{4'b1010}} repite dos veces la secuencia 1010.

Una constante en complemento dos se anota con el signo precediendo la secuencia: -4d'6 define el complemento dos de 6 en un largo de 4 bits.

El nombre de una variable puede tener componentes. En este caso se define un **arreglo**.

```
reg datos[31:0];    // define a datos como un arreglo de 32 bits.
Puede tenerse acceso individual a los bits, por ejemplo: datos[5]=1'bz;
Notar que en la definición del arreglo los corchetes van después del nombre del arreglo.
```

Una **memoria** es un arreglo de registros.

```
reg [15:0] mem16_1024 [1023:0];
```

Ejemplo A5.12. Mux con dos buses de entrada.

El esquema en base a lenguajes permite generalizaciones.

Una de las generalizaciones más comunes es cuando en lugar de tener una señal se tienen varias con funcionalidad similar. Puede efectuarse generalizaciones relativas al ancho del bus.

El siguiente módulo modifica el del Ejemplo A5.12, cambiando las entradas y salidas, de un alambre, por un bus de 8 señales.

```
// Multiplexer nivel proceso abstracto. Multiplexer 2 vías a una de 8 bits cada una.
// Cuando c=1, out=x1; y cuando c=0, out=x0

module mux21_buses (out, c, x1, x0);

    output [7:0] out;    // mux output de 8 bit
    input  c;
    input [7:0] x1, x0; // mux inputs de 8 bits cada una

    wire c;
    wire [7:0] x1, x0;
    reg [7:0] out;

    always @(*)
        if (c==1) out=x1; else out=x0;

endmodule
```

Debe notarse que una descripción estructural, en base a compuertas, sería muy costosa en términos de representación. Nótese que el segmento anterior puede ser generalizado a buses de mayor ancho.

Para esto puede emplearse una declaración de constantes:

```
parameter <constante>=<valor>;
```

El siguiente es un ejemplo de uso:

```
parameter ancho=16;
output [ancho-1: 0] out;    //bus de 16 bits
```

Ejemplo A5.13. Mux con varios buses de entrada.

Puede generalizarse aún más el concepto de multiplexor, cambiando el número de vías de entrada. Se diseña un mux de 4 buses de entrada con 8 bits cada uno.

```
// Multiplexer nivel proceso abstracto. 4-1 multiplexer, buses de 8 bits.
// Cuando c=3, out=x3; cuando c=2, out=x2; cuando c=1, out=x1; cuando c=0, out=x0

module mux4_1 (out, c, x3, x2, x1, x0);
    output [7:0] out;      // mux output
    input  [1:0] c;
    input  [7:0] x3, x2, x1, x0; // mux inputs

    wire [1:0] c;
    wire [7:0] x3, x2, x1, x0;
    reg  [7:0] out;

    always @(*)
        if (c==3) out=x3;
        else if (c==2) out=x2;
        else if (c==1) out =x1;
        else out= x0;
endmodule
```

Cada vez más la descripción de la funcionalidad o del comportamiento se va alejando de las tradicionales representaciones de sistemas digitales mediante compuertas o esquemáticos.

Adicionalmente puede mencionarse que la entrada de diseños en base a una biblioteca de esquemáticos de componentes tiene sus limitaciones. Es posible que vengan bloques de mux, prediseñados de un ancho fijo; sin embargo si un diseñador requiere uno de otro ancho, tendrá dificultades; en diseños mediante lenguajes, para cambios como los que mencionamos, sólo basta cambiar una constante.

El proceso también podría describirse mediante una sentencia case.

Ejemplo A5.14. Mux con operador condicional.

El siguiente ejemplo muestra un ejemplo de uso de sentencia condicional.

```
module mux4to1 (input x0, x1, x2, x3, c1, c0, output out);
    assign out = c1 ? (c0 ? x3 : x2) : (c0 ? x1 : x0);
endmodule
```




Figura A5.12. mux4to1.

Nótese que el módulo podría haberse desarrollado mediante vectores:

```

module mux4to1 (X, c, out);
    input [0:3] X;
    input [1:0] c;
    output out;
    reg out;

    always @(X or c)
        if (c == 0)    out = X[0];
        else if (c == 1) out = X[1];
        else if (c == 2) out = X[2];
        else if (c == 3) out = X[3];

endmodule
  
```

Dibujar un esquemático, empleando como componente al mux de 4 vías a una, es una buena manera de representar la arquitectura de un mux de 16 vías a una.

El diagrama ayuda a escribir el módulo de un mux de 16 vías a 1, en base a cinco muxes de 4 vías a una, mediante el empleo de vectores, puede escribirse:

```

module mux16to1 (X, C, out);
    input [0:15] X;
    input [3:0] C;
    output out;
    wire [0:3] n;    //nodos

    mux4to1 Mux1 (X[0:3],    C[1:0], n[0]);
    mux4to1 Mux2 (X[4:7],    C[1:0], n[1]);
    mux4to1 Mux3 (X[8:11],   C[1:0], n[2]);
    mux4to1 Mux4 (X[12:15],  C[1:0], n[3]);
    mux4to1 Mux5 (n[0:3],    C[3:2], out);

endmodule
  
```

Ejemplo A5.15. ALU con operandos vectores.

```
// 74381 ALU
module alu(s, A, B, F);
    input [2:0] s;
    input [3:0] A, B;
    output [3:0] F;
    reg [3:0] F;

    always @(s or A or B)
        case (s)
            0: F = 4'b0000;
            1: F = B - A;
            2: F = A - B;
            3: F = A + B;
            4: F = A ^ B;
            5: F = A | B;
            6: F = A & B;
            7: F = 4'b1111;
        endcase
endmodule
```

Ejemplo A5.16. Sumador completo con operandos vectores.

Empleando vectores se reduce apreciablemente la lista de argumentos en las puertas del módulo. Por ejemplo el módulo que representa a un sumador completo de 4 bits resulta:

```
module fulladder4 (Cin, a, b, sum, Cout);
    input carryin;
    input [3:0] a, b;
    output [3:0] sum;
    output carryout;
    wire [3:1] c;

    sumadorcompleto etapa0 (Cin, a[0], b[0], sum[0], c[1]);
    sumadorcompleto etapa1 (c[1], a[1], b[1], sum[1], c[2]);
    sumadorcompleto etapa2 (c[2], a[2], b[2], sum[2], c[3]);
    sumadorcompleto etapa3 (c[3], a[3], b[3], sum[3], Cout);

endmodule
```

Comparar con A5.4.

Ejemplo A5.17. Estructuras repetitivas. Lazos for dentro de procesos.

Una de las ventajas de las descripciones mediante lenguajes, es que las estructuras que son repetitivas pueden ser planteadas dentro de un lazo for dentro de un proceso.

```

module AdderBitSlice (Cin, a, b, sum, Cout);
    parameter n=32;          //n es una constante
    input Cin;
    input [n-1:0] a, b;
    output [n-1:0] sum;
    output Cout;

    // Las señales que son escritas dentro de un procedimiento deben ser declaradas reg.
    reg [n-1:0] sum;
    reg Cout;
    reg [n:0] c;              //reservas intermódulos
    integer k;                //variable entera para índice de for

    always @(a or b or Cin)
    begin
        c[0] = Cin;
        for (k = 0; k < n; k = k+1)
            begin
                sum[k] = a[k] ^ b[k] ^ c[k];
                c[k+1] = (a[k] & b[k]) | (a[k] & c[k]) | (b[k] & c[k]);
            end
        Cout = c[n];
    end

endmodule

```

La compilación o la traducción a compuertas del dispositivo programable puede que de origen a un tiempo considerable para efectuar la suma. Una de las soluciones es cambiar la arquitectura de la propagación ondulada de la reserva por algún esquema de generación adelantada y agrupada de las reservas. Esto también puede describirse mediante lenguajes.

Ejemplo A5.18. Descripción algorítmica, empleando operadores del lenguaje.

Empleando el operador que realiza la suma puede describirse un sumador binario en forma muy compacta. La operación de concatenación de registros permite generar Cout correctamente. La señal overflow considera operandos con signo.

```

module addern (Cin, a, b, sum, Cout, overflow);
    parameter n = 32;
    input Cin;
    input [n-1:0] a, b;
    output [n-1:0] sum;
    output Cout, overflow;
    reg [n-1:0] sum;
    reg Cout, overflow;

```

```

always @(a or b or Cin)
begin
    { Cout, sum } = a + b + Cin;    //descripción funcional
    overflow = Cout ^ a[n-1] ^ b[n-1] ^ sum[n-1];
end

endmodule

```

Las dos últimas columnas de la tabla de la Figura A5.13, muestran la generación de la señal overflow, sin emplear la entrada C31. Se ilustra la señal LT (less than) para operandos con signo.

A31	B31	C31	Co	S31	Ov	LT	A31^B31	C0^S31
0	0	0	0	0	0	0	0	0
0	0	1	0	1	1	0	0	1
0	1	0	0	1	0	1	1	1
0	1	1	1	0	0	0	1	1
1	0	0	0	1	0	1	1	1
1	0	1	1	0	0	0	1	1
1	1	0	1	0	1	1	0	1
1	1	1	1	1	0	1	0	0

Figura A5.13.

Ejemplo A5.19. Decodificador, empleando operadores del lenguaje.

Un decodificador de 2 líneas a 4, cuyas señales se muestran en el diagrama en bloques de la Figura A5.14, puede describirse funcionalmente según:

```

module dec2a4 (x, y, Enable);
    input [1:0] x;
    input Enable;
    output [0:3] y;
    reg [0:3] y;
    integer k;

    always @(x or Enable)
        for (k = 0; k <= 3; k = k+1)
            if ((x == k) && (Enable == 1))
                y[k] = 1;
            else
                y[k] = 0;
endmodule

```





Figura A5.14.

Para la siguiente descripción jerárquica, de un decodificador de 4 líneas a 16, en base al módulo anterior, resulta útil desarrollar la arquitectura; es decir, un esquema de las conexiones entre los cinco bloques básicos, antes de escribir el código.

```

module dec4to16 (x, y, Enable);
    input [3:0] x;
    input Enable;
    output [0:15] y;
    wire [0:3] nodosenables;

    dec2to4 Dec2 (x[1:0], y[0:3],    nodosenables [0]);
    dec2to4 Dec3 (x[1:0], y[4:7],    nodosenables [1]);
    dec2to4 Dec4 (x[1:0], y[8:11],   nodosenables [2]);
    dec2to4 Dec5 (x[1:0], y[12:15],  nodosenables [3]);
    dec2to4 Dec1 (x[3:2], nodosenables [0:3], Enable);

endmodule
  
```

Ejemplo A5.20. Buffer de tercer estado

La Figura A5.15 muestra un buffer de tercer estado, de 8 bits.

```

module tristate (X, OE, Y);
    parameter n = 8;
    input [n-1:0] X;
    input OE;
    output [n-1:0] Y;
    wire [n-1:0] Y;

    assign Y = OE ? X : 'bz;

endmodule
  
```




Figura A5.15.

Ejemplo A5.21. Relación entre simulación y síntesis.

El siguiente segmento ilustra una situación de síntesis no combinacional. Esto se advierte ya que W y Z no son asignadas en cada una de las alternativas de control. Por ejemplo cuando A es 1, se cambia Z al valor de B; pero si A es 0, Z no cambia. Similar situación se tiene para W. Esta descripción plantea que en caso de ser A igual a 0, el sistema recuerda el valor anterior de Z. Esta conducta no es de tipo combinacional, ya que los sistemas combinacionales no recuerdan valores previos. En esta situación al efectuar la síntesis se emplearán flip-flops para las variables Z y W.

```
input A, B;
output W, Z;
reg W, Z;
always @(A or B )
begin
    if (A==1) Z=B;
    else W=A;
end
```

En el caso del ejemplo, es preferible emplear $Z \leq B$ y $W \leq A$, para las asignaciones, de esta forma la simulación y la síntesis darán iguales resultados.

En el proceso de **simulación**, del Ejemplo A5.9, se considera a Z y W como variables combinacionales, pero en el proceso de **síntesis** se las considerará secuenciales.

A5.7 Sistemas secuenciales.**Latch transparente. (D gated latch)**

La Figura A5.17 muestra las formas de ondas de un latch transparente, su principal característica, es que desde el canto de subida del reloj se copia la entrada hacia la salida (este es el modo transparente); y cuando ocurre el canto de bajada del reloj el flip-flop la salida queda capturada.

Un esquemático basado en compuertas, se muestra en la Figura A5.16.




Figura A5.16.

Un módulo que describe la estructura usando compuertas, es el siguiente:

```
//latch con compuertas.
module latchc(input D, clk, output Q);
  wire P, S, R;

  and #1 (S, D, clk);
  not #1 (P, clk);
  and #1 (R,P,Q);
  or #2 (Q, S, R); //retardo mayor es necesario para correcta simulación

endmodule
```

Para simular correctamente circuitos con compuertas realimentadas, es necesario definirles el retardo asociado a cada una de ellas.

La Figura A5.20 describe funcionalmente el latch transparente, empleando un multiplexor.

```
//latch proceso.
module latchp(input D, clk, output Q);
  reg Q;
  always @(D or clk)
    if (clk==1) Q = D;
    else Q = Q;
endmodule
```




Figura A5.17.

Note que las asignaciones son bloqueantes, ya que se está representando un sistema cuya salida puede cambiar entre cantos del reloj.

No importando la arquitectura, un símbolo lógico del latch transparente se muestra en la Figura A5.18.




Figura A5.18.

La tabla característica del latch, que muestra el próximo estado en función de las entradas, se muestra en la Figura A5.19.

clk	D	Q(k+1)
0	x	Q(k)
1	0	0
1	1	1

Figura A5.19.

La Figura A5.20, muestra diferentes casos de secuencias de las entradas D y clk. Cuando el reloj tiene un canto de subida se deja pasar la entrada hacia la salida, cuando el reloj está bajo los cambios de la entrada no afectan al latch. Si ocurren cambios de la entrada con el reloj en alto, éstos son considerados.

Note en el intervalo t_3 , la caída de Q cuando baja D. También debe advertirse que en el intervalo t_4 , estando el reloj bajo, un pulso en D no alcanza a ser registrado por el latch. Luego de t_4 , estando el reloj alto, un canto de subida de D es capturado por el latch.




Figura A5.20.

El siguiente módulo genera estímulos y despliega los resultados al ser aplicados los estímulos de la Figura A5.20 a las implementaciones del latch transparente basadas en compuertas y en un mux.

```
module testlatch(input Q, output clk, D);
```

```

reg clk, D;
initial      //proceso que permite grabar en archivo las formas de ondas.
begin
    $dumpfile("latcht.vcd");
    $dumpvars(0, $time, clk, D, Q);
end

initial      //proceso que monitorea los cambios.
begin
    $monitor("clk=%b, D=%b, Q=%b t=%d", clk, D, Q, $time);
    clk=0; D=1; //D=1 para iniciar latch con compuertas.
    // clk=0; D=0; //D=0 para iniciar latch basado en mux.
    #1 clk=1;
    #1 clk=0;
    #1 D=0; //hasta aquí son inicializaciones para simulación.
    #10 D=1;
    #10 clk=1;
    #20 clk=0;
    #10 D=0;
    #20 clk=1;
    #20 clk=0;
    #10 D=1;
    #20 clk=1;
    #10 D=0;
    #10 clk=1;
    #20 clk=0;
    #10 D=1;
    #10 D=0;
    #10 clk=1;
    #10 D=1;
    #10 clk=0;
    #20 D=0;
    #10 $finish;
end
endmodule

module BancoDePrueba;
wire D, clk, Q, co;
    latchc  l1(D, clk, Q);
// latchp  l1(D, clk, Q);
    testlatch t1(Q, clk, D);
endmodule

```

Para el latch basado en compuertas, puede obtenerse la siguiente salida, empleando un simulador verilog, de los varios disponibles como software libre en la red.

```

clk=0, D=1, Q=x t=          0 //Al inicio Q es desconocido

```

clk=1, D=1, Q=x t=	1	
clk=0, D=1, Q=x t=	2	
clk=0, D=0, Q=x t=	3	//Aquí puede considerarse iniciada la simulación.
clk=0, D=0, Q=0 t=	4	
clk=0, D=0, Q=0 t=	5	
clk=0, D=1, Q=0 t=	13	
clk=1, D=1, Q=0 t=	23	
clk=1, D=1, Q=1 t=	26	
clk=0, D=1, Q=1 t=	43	
clk=0, D=1, Q=1 t=	46	
clk=0, D=1, Q=1 t=	47	
clk=0, D=0, Q=1 t=	53	
clk=1, D=0, Q=1 t=	73	
clk=1, D=0, Q=0 t=	77	
clk=0, D=0, Q=0 t=	93	
clk=0, D=1, Q=0 t=	103	
clk=1, D=1, Q=0 t=	123	
clk=1, D=1, Q=1 t=	126	
clk=1, D=0, Q=1 t=	133	
clk=1, D=0, Q=0 t=	136	
clk=0, D=0, Q=0 t=	163	
clk=0, D=1, Q=0 t=	173	
clk=0, D=0, Q=0 t=	183	
clk=1, D=0, Q=0 t=	193	
clk=1, D=1, Q=0 t=	203	
clk=1, D=1, Q=1 t=	206	
clk=0, D=1, Q=1 t=	213	
clk=0, D=1, Q=1 t=	216	
clk=0, D=1, Q=1 t=	217	
clk=0, D=0, Q=1 t=	233	

Figura A5.21.

Para el basado en mux, puede obtenerse una visualización de las formas de ondas, a partir de un archivo en formato estándar vcd; existen aplicaciones de software libre para visualizar estos archivos.




Figura A5.22.

Flip-flop D. Disparado por canto de subida.

La Figura A5.23 muestra un flip-flop D disparado por cantos de subida del reloj. El módulo en Verilog, emplea la condición posedge para definir el evento canto de subida. La asignación a Q es no bloqueante.

El nombre de la función posedge recuerda **positive edge** (canto de subida, en español).

```
module flipflop (input D, clk, output Q);
    reg Q;
    always @(posedge clk)
        Q <= D;
endmodule
```




Figura A5.23.

La Figura A5.24 muestra el funcionamiento, ahora la señal Q permanece constante entre ciclos de reloj. La salida registra el valor de la entrada en el momento de ocurrir el canto. En un dispositivo real es preciso que no ocurran cambios de la entrada un intervalo antes del canto de subida del reloj (set-up); y un intervalo después del canto (hold).




Figura A5.24.

Flip-flop D. Disparado por canto de bajada.

En la descripción Verilog, basta cambiar el evento a negedge. En el símbolo lógico, de la Figura A5.25, se coloca un círculo que simboliza un inversor en la entrada del reloj.

```
module flipflop (input D, clk, output Q);
    reg Q;
    always @(negedge clk)
        Q <= D;
endmodule
```




Figura A5.25.

Flip-flop T.

T	Q(k+1)
0	Q(k)
1	Q'(k)

Figura A5.26.




Figura A5.27.

```
module tff(input T, clk, output Q);
    reg q;
    always @ (posedge clk)
        if(T) Q<=~Q; else Q<= Q;
endmodule
```

Controles de reset asincrónicos y sincrónicos.

Los siguientes segmentos ilustran un flip-flop D, con controles de reset sincrónico y asincrónico.

Si en el proceso (always) que describe al flip-flop D, la señal reset no está en la lista de sensibilidad, sólo se activarán los cambios de la salida cuando se produzca un canto de subida del reloj. Entonces el reset se producirá en un canto del reloj, por esto se dice que es sincrónico.

```
module flipflop (input D, clk, reset, output Q);
    reg Q;
    always @(posedge clk)
        if (reset) Q <= 0;    //reset sincrónico
        else Q <= D;
endmodule
```




Figura A5.28.

El siguiente proceso produce el reset asincrónico del flip-flop D. Ya que se producirá el cambio de estado siempre que cambie la señal reset, que puede ser asincrónica con el reloj.

```

module flipflop (input D, clk, reset, output Q);
    reg Q;
    always @(posedge clk or posedge reset)
        if (reset) Q = 0;    //reset asincrónico
        else Q <= D;
endmodule

```

El reset asincrónico se asigna en forma bloqueante, de esta manera la salida irá a cero apenas reset suba, incluso entre cantos de reloj.

En el ejemplo se emplean asignaciones bloqueantes y nobloqueantes a una misma señal; sin embargo no es recomendable mezclarlas en el mismo proceso o bloque always.

Registro *n* bits.

La descripción de la arquitectura de un registro con habilitación de escritura, teniendo como parámetro del ancho del registro, puede escribirse:

```

module registro (R, WE, Clk, Q);
    parameter n = 8;
    input [n-1:0] R;
    input WE, Clk;
    output [n-1:0] Q;
    reg [n-1:0] Q;

    always @(posedge Clk)
        if (WE) Q <= R;

endmodule

```

Un diagrama lógico del registro se muestra en la Figura A5.29.




Figura A5.29.

Carreras en simulación.

No es recomendable efectuar asignaciones a la misma variable en dos procesos diferentes, con el fin de evitar carreras.

Como se define que en la simulación que realiza Verilog, los procesos pueden ser itinerados en cualquier orden. Luego de un pulso de reset, las variables y1 e y2 podrían estar ambas en cero o ambas en uno. Situación que define una carrera.

```
reg y1, y2;
```

```
always @(posedge clk or posedge reset)
if (reset) y1 = 0; // reset
else y1 = y2;
```

```
always @(posedge clk or posedge reset)
if (reset) y2 = 1; // preset
else y2 = y1;
```

Si las asignaciones, dentro de los procesos, se cambian a nobloqueantes, luego de un reset, y1 quedará en cero e y2 en 1.

Registro de desplazamiento.

La descripción de la arquitectura de un registro de desplazamiento lógico hacia la derecha de 4 bits, con la función adicional de carga paralela, puede plantearse:

```
module shift4 (C, Load, x, Clk, Q);
  input [3:0] C; //carga paralela
  input Load, x, Clk;
  output [3:0] Q;
  reg [3:0] Q;
  always @(posedge Clk)
    if (Load) Q <= C; //carga sincrónica
    else
      begin
        Q[0] <= Q[1]; Q[1] <= Q[2];
        Q[2] <= Q[3]; Q[3] <= x;
      end
endmodule
```

Debido a que las asignaciones son nobloqueantes, no importa el orden en que están escritos los corrimientos individuales de los bits.




Figura A5.30.

El esquema anterior permite generalizar a un registro de ancho n , empleando una sentencia `for`.

```
//Registro de desplazamiento ancho n.
module shiftn (R, Load, x, Clk, Q);
    parameter n = 16;
    input [n-1:0] C;
    input Load, x, Clk;
    output [n-1:0] Q;
    reg [n-1:0] Q;
    integer k;

    always @(posedge Clk)
        if (Load) Q <= C;
        else
            begin
                for (k = 0; k < n-1; k = k+1) Q[k] <= Q[k+1];
                Q[n-1] <= x;
            end
endmodule
```

Una descripción del corrimiento a la derecha empleando la operación concatenación es:
 $\{Q, Q[0]\} = \{x, Q\};$

Contador ascendente con reset de lógica negativa.

Una señal de lógica negativa se considera activada por cantos de bajada. En los diagramas lógicos suele denominarse en forma complementada, y su lugar de entrada al bloque pasa por un pequeño círculo. En la Figura A5.31 se muestra la señal con nombre complementado.

```
module upcounter (input reset, Clk, Enable, output [3:0] Q);
    reg [3:0] Q;
```

```

always @(posedge reset or posedge Clk)
    if (!reset) Q <= 0;
    else if (Enable) Q <= Q + 1;
endmodule

```




Figura A5.31.

Contador con reset, carga paralela y habilitación para contar.

La Figura A5.32 muestra el contador con sus funciones.

```

module upcount (C, reset, Clock, Enable, Load, Q);
    input [3:0] C;
    input reset, Clk, Enable, Load;
    output [3:0] Q;
    reg [3:0] Q;

    always @(posedge reset or posedge Clk)
        if (!reset) Q <= 0;
        else if (Load)
            Q <= C;
        else if (Enable)
            Q <= Q + 1;
endmodule

```




Figura A5.32.

Divisor de frecuencia por n.

En un contador interno Cnt va contando, desde 0 hasta (Divisor-1) veces el reloj de alta frecuencia clk. Cuando cuenta el pulso Divisor mantiene durante un pulso, del reloj de alta frecuencia, un uno en el reloj de baja frecuencia; en el resto de las cuentas de cnt el reloj de baja frecuencia permanece en cero.

```
module divisor(input Clk, reset, output Clkd);
    parameter Divisor = 4;
    parameter Bitcnt = 1; // 2 elevado a (Bitcnt+1) debe ser >= Divisor
    reg    Clkd;
    reg    [Bitcnt:0] Cnt; //debe poder contar desde 1 a Divisor.
```

```
    always @ (posedge Clk or negedge reset)
        if(!reset) begin Clkd <= 0; Cnt <= 0; end
        else if (Cnt == Divisor-1) begin Clkd <= 1; Cnt <=0; end
        else begin Clkd <= 0; Cnt <= Cnt+1; end
```

```
endmodule
```

```
module testdivisor(input clkd, output clk, reset);
    reg clk, reset;
    integer k;
    initial
    begin
        $dumpfile("divisor.vcd");
        $dumpvars(0, $time, reset, clk, clkd);
    end
```

```
    initial begin: DetenerSimulacion
        // Detiene la simulación después de 100 unidades de tiempo de simulación.
        //Es necesaria si se emplea simulador de reloj
        #100 $stop;
    end
```

```
    initial
    begin
        $monitor("reset=%b, clk=%b, clkd=%b t=%d", reset, clk, clkd, $time);
        reset=0; clk=0;
        #5 reset=1;
        //alternativa para generar una cantidad finita de pulsos de reloj.
        //for(k=0;k<30;k=k+1) #10 clk<=~clk; $finish;    End
```

```
    always @(clk) #10 clk <=~clk;    //genera infinitos pulsos
endmodule
```

```
module BancoDePrueba;
```

```

wire reset, clk, clkd;
divisor d1(clk, reset, clkd);
testdivisor t1(clkd, clk, reset);
endmodule

```

La Figura A5.33 muestra el período del reloj de baja frecuencia, se cumplen las relaciones, con T el período del reloj de alta frecuencia.

$$Clkd = \frac{Clk}{Divisor}$$

$$Td = Divisor \cdot T$$




Figura A5.33.

Las formas de ondas fueron generadas con el archivo divisor.vcd, producido en la simulación.

El siguiente módulo emplea una asignación bloqueante para actualizar la señal clk . Este cambio se produce dentro del paso de simulación y no generará nuevos eventos; por lo tanto no oscila. La construcción correcta es con una asignación nobloqueante.

```

module oscilador (output clk);
reg clk;
initial #10 clk = 0;
always @(clk) #10 clk = ~clk; //debe cambiarse a: clk <= ~clk;
endmodule

```

Transferencias entre registros de segmentación en pipeline.

En las transferencias entre registros en secuencia, de la Figura A5.34, deben emplearse asignaciones no bloqueantes. De este modo no importa el orden de las asignaciones y además cada una de las asignaciones podría estar en procesos `always` diferentes. Con esta elección, que refleja la naturaleza secuencial del proceso, la simulación y la síntesis dan iguales resultados.




Figura A5.34.

```

module segmentos (q3, d, clk);
output [7:0] q3;
input [7:0] d;
input clk;
reg [7:0] q3, q2, q1;
always @(negedge clk) begin
q3 <= q2;
q2 <= q1;
q1 <= d;
end
endmodule

```

Registro de desplazamiento con realimentaciones lineales.

Un Linear Feedback Shift-Register (LFSR) es una máquina secuencial que posee realimentaciones. Uno de sus principales usos es en encriptación.




Figura A5.35.

```

module lfsr (output q2, input clk, presetn);
reg q2, q1, q0;
wire nodo;
assign nodo = q0 ^ q2;
always @(posedge clk or negedge presetn)
if (!presetn) begin q2 <= 1'b1; q1 <= 1'b1; q0 <= 1'b1; end
else begin q2 <= q1; q1 <= nodo; q0 <= q2; end

```



```
endmodule
```

Una alternativa para el cuerpo del proceso es emplear la construcción de concatenación.

```
always @(posedge clk or negedge presetn)
if (!presetn) {q2,q1,q0} <= 3'b111; else {q2,q1,q0} <= {q1,(q0^q2),q2};
```

El siguiente diseño considera un registro con carga paralela y notación vectorial.

```
module lfsr3(c, Load, clk, q);
    input [0:2] c;
    input Load, clk;
    output [0:2] q;
    reg [0:2] q;

    always @ (posedge clk)
        if(Load) q <= c; else q <= {q[2], q[0]^q[2], q[1]};
endmodule
```

Los siguientes módulos simulan el funcionamiento del módulo anterior.

```
module testlfsr(q, clk, Load, carga);
input [0:2] q;
output clk;
output Load;
output [0:2] carga;

reg clk, Load;
reg [0:2] carga;

initial
begin
    $dumpfile("lfsr3.vcd");
    $dumpvars(0, $time, Load,carga, clk, q);
end

initial begin: DetenerSimulacion
// Detiene la simulación después de 100 unidades de tiempo de simulación.
#400 $stop;
end

initial
begin
    $monitor("Load=%b carga=%b clk=%b salida=%b t=%d", Load, carga, clk, q, $time);
    Load=0; clk=0;carga=0;
    #10 Load=1;
    #10 carga=1;
```

```

    #10 Load=0;
    #10 carga=0;
end
always @(clk) #5 clk <=~clk;
endmodule

```

```

module BancoDePrueba;
wire Load, clk;
wire [0:2] c;
wire [0:2] q;

lfsr3 lf1(c, Load, clk, q);
testlfsr t1(q, clk, Load, c);
endmodule

```

Se obtienen las siguientes formas de ondas:




Figura A5.36.

Contador de anillo

Se realimenta un registro de desplazamiento a la izquierda, formando un anillo. Se inicia el bit menos significativo en 1, y se lo va desplazando. Las cuentas son potencias de dos. El uso de la operación concatenación, permite describir la arquitectura en función de n , el número de bits del registro.

La señal $reset'$, al activarse coloca en 1 el flip-flop q_0 , mediante la señal S (set); y en cero el resto de los flip-flops, mediante la señal C (clear).




Figura A5.37

```

module cntanillo(resetn, clk, Q);
    parameter n = 8;
    input resetn, clk;
    output [n-1:0] Q;
    reg [n-1:0] Q;
    always @(posedge clk)
        if (!resetn) begin Q[n-1:1] <= 0; Q[0] <= 1; end
        else Q <= {{ Q[n-2:0]}, { Q[n-1] }};
endmodule

```

Contador BCD de dos cifras.

El módulo siguiente contador BCD de dos cifras, módulo 100.
 Con funciones de Clear, para dejar en cero y Enable para habilitar las cuentas.
 La función Clear opera sincrónica con el reloj.

```

module BCDcount (Clk, Clear, Enable, BCD1, BCD0);
    input Clk, Clear, Enable;
    output [3:0] BCD1, BCD0;
    reg [3:0] BCD1, BCD0;

    always @(posedge Clk)
    begin
        if (Clear) begin BCD1 <= 0; BCD0 <= 0; end
        else
            if (Enable)
                if (BCD0 == 4'b1001)
                    begin //rebalse unidades
                        BCD0 <= 0;
                        if (BCD1 == 4'b1001) BCD1 <= 0; //rebalse decenas, módulo100.
                        else BCD1 <= BCD1 + 1;
                    end
                else BCD0 <= BCD0 + 1;
            end
    end
endmodule

```

Ejemplos de descripciones de comportamiento.

```

module GeneraCondicion(gt, A, B);
    parameter n=4;
    output gt;
    input [n-1:0] A, B;
    assign gt = (A > B);
endmodule

```

```

module ObtieneElMayor (mayor, A, B);
    parameter n=4;
    output [n-1:0] mayor;

```

```

input [n-1:0] A, B;
  assign mayor = (A > B) ? A : B;
endmodule

```

A5.8. Máquinas secuenciales de estados finitos.

Las herramientas de apoyo al diseño permiten ingresar un diagrama de estados en forma gráfica; el cual luego es traducido a un programa Verilog.

Una alternativa es describir directamente mediante lenguajes. La situación puede plantearse mediante dos o tres procesos.

Diagramas de Moore.

Para una máquina de Moore, se tendrán tres procesos, los cuales se describen en la Figura A5.38.




Figura A5.38

El diagrama de estados, de la Figura A5.39, describe una máquina que cuando detecta dos o más unos seguidos en la entrada x entrega salida $z=1$; en caso contrario la salida será $z=0$.




Figura A5.39

El siguiente módulo describe el diagrama de la Figura A5.39, siguiendo el esquema de la Figura A5.38.

```
//Detecta dos o más unos seguidos de la entrada x con salida z=1.
module moore (input Clk, Resetn, x, output z);
    reg [1:0] y, Y;
    parameter [1:0] A = 2'b00, B = 2'b01, C = 2'b10;
    //Asignación de estados. Codificación binaria

    // Red combinacional de próximo estado. P1
    always @(x or y)
        case (y)
            A: if (x) Y = B; else Y = A;
            B: if (x) Y = C; else Y = A;
            C: if (x) Y = C; else Y = A;
            default: Y = 2'bxx;
        endcase
    // Proceso secuencial del registro. P2
    always @(negedge Resetn or posedge Clk)
        if (Resetn == 0) y <= A;
        else y <= Y; //estado presente<=próximo estado

    // Bloque combinacional de salida. P3
    assign z = (y == C);
endmodule
```

La generación de estímulos en máquinas secuenciales es difícil. En casos sencillos es posible generar todas las secuencias que recorran las diferentes opciones que da el diagrama. Existen herramientas para generar estímulos, en forma gráfica e interactiva. Como ejemplo se muestran algunos estímulos para revisar el funcionamiento, mediante simulación.

```
module test(output Clk, Resetn, x, input z);
reg Clk, Resetn, x;
initial
begin
    $monitor("clk=%b reset=%b x=%b z=%d", Clk, Resetn, x, z, $time);
    Clk=0; Resetn=1; x=0;
    #10 Clk=1; Resetn=0;
    #10 Clk=0; Resetn=1;

    #10 Clk=0;x=1;//un pulso de reloj => tres vectores.
    #10 Clk=1;x=1;//Cambios de señal con reloj bajo.
    #10 Clk=0;x=1;

    #10 Clk=0;x=0;//un pulso de reloj => tres vectores.
    #10 Clk=1;x=0;
    #10 Clk=0;x=0;
```

```

#10 Clk=0;x=1;//un pulso de reloj => tres vectores.
#10 Clk=1;x=1;
#10 Clk=0;x=1;
#10 Clk=0;x=1;//un pulso de reloj => tres vectores.
#10 Clk=1;x=1;
#10 Clk=0;x=1;

#10 Clk=0;x=0;//un pulso de reloj => tres vectores.
#10 Clk=1;x=0;
#10 Clk=0;x=0;
#10 $finish;
end
endmodule

module BancoDePrueba;
wire Clk, Resetn, x, z;
  moore m1 (Clk, Resetn, x, z);
  test t1 (Clk, Resetn, x, z);
endmodule

```

Diagramas de Mealy.

El programa Verilog, está basado en dos procesos, P1 es combinacional y P2 secuencial. Note que P2 es igual al de la máquina similar de Moore.




Figura A5.39

El diagrama de estados de Mealy, se muestra en la Figura A5.40, con un estado menos que el similar de Moore.




Figura A5.40

//Detecta dos o más unos seguidos de la entrada x con salida z=1.

module mealy (**input** Clk, Resetn, x, **output** z);

reg y, Y, z;

parameter A = 0, B = 1;

// Redes combinatoriales de próximo estado y de salida. P1

always @(x or y)

case (y)

A: **if** (x) **begin** z = 0; Y = B; **end**

else begin z = 0; Y = A; **end**

B: **if** (x) **begin** z = 1; Y = B; **end**

else begin z = 0; Y = A; **end**

endcase

// Proceso registro. P2

always @(negedge Resetn or posedge Clk)

if (Resetn == 0) y <= A;

else y <= Y;

endmodule

Ejemplo.

Compara las entradas x1 e x2, si durante cuatro pulsos de reloj, se tienen secuencias iguales se produce salida z=1; en caso contrario la salida es cero.

Para las siguientes secuencias, se tiene la salida:

x1: 0 1 1 0 1 1 1 0 0 0 1 1 0

x2: 1 1 1 0 1 0 1 0 0 0 1 1 1

z: 0 0 0 0 1 0 0 0 0 1 1 1 0

module ejemploMealy(input Clk, Resetn, x1, x2, output z);

reg z;

reg [1:0] y, Y;

wire x;

```

parameter [1:0] A = 2'b00, B = 2'b01, C = 2'b10, D = 2'b11;
// Define the next state and output combinational circuits
assign x = x1 ^ x2;
always @(x or y)
    case (y)
        A: if (x) begin Y = A; z = 0; end
           else begin Y = B; z = 0; end
        B: if (x) begin Y = A; z = 0; end
           else begin Y = C; z = 0; end
        C: if (x) begin Y = A; z = 0; end
           else begin Y = D; z = 0; end
        D: if (x) begin Y = A; z = 0; end
           else begin Y = D; z = 1; end
    endcase

// Define the sequential block
always @(negedge Resetn or posedge Clk)
    if (Resetn == 0) y <= A;
    else y <= Y;
endmodule

```

A5.9. Sistemas asincrónicos.

Latch de nand.

```

module latchnand(q, qprima, setn, resetn);
output q, qprima;
input setn, resetn;

nand #1 G1(q, qprima, setn);
nand #1 G2 (qprima, q, resetn);

endmodule

```




Figura A5.41

La siguiente serie de estímulos genera diferentes recorridos. Si ambos controles pasan por cero, no puede asegurarse el estado del latch.


```

initial
begin
    $monitor("s=%b r=%b q=%b q'=%b", setn, resetn, q, qprima, $time);
    setn=1;resetn=1;
    #2 setn=0;
    #2 setn=1;
    #4 setn=0;
    #4 setn=1;
    #12 resetn=0;
    #4 resetn=1;
    #10 setn=0;resetn=0;
    #4 setn=1;
    #4 setn=0;
    #4 setn=1;
    #4 setn=0;
    $finish;

end

```

Descripción estructural flip-flop D, en base a nand.

```

//Descripción conducta flip-flop D
module dffc(output q, input clk, d);
    reg q;
    always @(negedge clk) q = #10 d;
endmodule

```

```

//Descripción estructural flip-flop D
module dffe(output q, input clk, d);
    wire q, qprima, r, s, r1, s1;
    nor G1 (q, qprima, r);
    nor G2 (qprima, q, s);
    nor G3 (s, r, clk, s1);
    nor G4 (s1, s, d);
    nor G5 (r, r1, clk);
    nor G6 (r1, s1, r);
endmodule

```




Figura A5.42


Flip-flop RS.

Figura A5.43

A5.10. Síntesis.

El proceso de síntesis describe las etapas para convertir, la descripción HDL (Verilog o VHDL) de un sistema digital, a una forma que pueda ser armada o manufacturada o descargada a un dispositivo programable.

El o los elementos en los que será implementado el diseño, se denomina sistema tecnológico objetivo (target), normalmente es de tecnología semiconductora. Se denomina ASIC (Application-Specific Integrated Circuit), a los dispositivos que se construyen especialmente para soportar un diseño particular. En los dispositivos programables como FPGA y CPLD el diseño es descargado al chip, que suele ser reprogramable.

Una metodología que puede emplearse es efectuar una descripción del comportamiento lo más abstracta posible, luego preparar una serie de estímulos que al ser aplicados al diseño, en una simulación, confirmen su funcionalidad. La preparación cuidadosa de un banco de pruebas suele ser más difícil que el diseño mismo.

Puede ser que algunas descripciones abstractas no puedan ser sintetizadas, ya que el programa de síntesis reconoce sólo las estructuras que se le han almacenado en su inteligencia artificial. Si por ejemplo puede reconocer: registros, sumadores, multiplexores, comparadores, contadores, etc., es porque se ha almacenado en bibliotecas la descripción (en Verilog) de estas unidades en términos de compuertas y flip-flops.

Si a pesar de lo abstracta de la descripción, ésta puede ser sintetizada, es decir traducida a la interconexión de compuertas y flip-flops, entonces se podría pasar a la siguiente etapa. En caso que esto no sea posible, el diseñador debe descomponer en partes su idea original, posiblemente mediante descripciones RTL de la arquitectura. Lo cual implica crear nuevos módulos, que deben ser sintetizables, cada uno de los cuales debe ser sometido a simulaciones funcionales cuidadosas.

Cuando ya se dispone de una red booleana, la siguiente etapa es minimizarla, normalmente en área, considerando que se tendrá multifunciones y multiniveles, esto es realizado por complejas heurísticas programadas en la herramienta CAD. Luego de esto viene el mapeo tecnológico, en el cual las configuraciones de compuertas y flip-flops se hacen calzar con los arreglos de multiplexores o tablas de búsqueda con que está formado el dispositivo tecnológico. Luego es necesario colocar e interconectar los bloques (en caso de dispositivos programables). En estas condiciones es preciso evaluar, mediante simulaciones temporales, si se cumplen los requerimientos de velocidad de los sistemas combinacionales y los tiempos de hold y set-up de los flip-flops. En caso de no cumplirse, deben rediseñarse partes y volver al lazo de diseño (ver Figura A5.1).

Mientras más abstracto el diseño, más rápido es el desarrollo.

Debido a la incorporación de nuevos algoritmos y heurísticas para efectuar la síntesis en términos de compuertas y flip-flops, debería evaluarse la necesidad de enfrentar un diseño en términos de más bajo nivel. Una forma de ir controlando el descenso de nivel, es llevar registro del número de elementos usados en el diseño y de los tiempos de ejecución, para cada una de las arquitecturas que se vayan desarrollando; esto permitirá evaluar si es preciso confeccionar módulos con más nivel de detalle. También ayuda, conocer los subsistemas que la herramienta de ayuda al diseño, que se está empleando, sintetiza eficientemente.

Una de las ventajas de Verilog, es que, por un lado, permite describir módulos en términos de compuertas y flip-flops, lo cual posibilita llegar a niveles muy bajos de abstracción. Por el otro extremo, a medida que se tienen más experiencias, se han desarrollado por expertos, algunas bibliotecas o núcleos que pueden ser reutilizados, entre ellos microcontroladores embebidos, lo cual facilita la realización de diseños abstractos sintetizables.

La forma más básica de programar o diseñar es mediante descripciones estructurales explícitas. Es decir construir módulos solamente con compuertas, sin emplear procesos (always,

initial) ni asignaciones continuas (assign). Son dificultosas de escribir pero fáciles y seguras de sintetizar.

Las descripciones estructurales implícitas, emplean los operadores al bit del lenguaje y el comando assign para describir expresiones. Tampoco se emplean procesos. Estas descripciones son más fáciles de escribir y también puede asegurarse que son sintetizables.

Algunos operadores más complejos (como la suma) suelen ser reconocidos por las herramientas de síntesis; el operador condicional también es sintetizable ya que suele traducirse a multiplexores.

En el mapeo tecnológico se traducen las compuertas genéricas (por ejemplo compuertas de 8 entradas) a los bloques lógicos del dispositivo; también algunas operaciones pueden ser mapeadas tecnológicamente (por ejemplo multiplicadores, si existen en el dispositivo), empleando este tipo de chips suele mapearse tecnológicamente con software que provee el mismo fabricante.

Una de las dificultades que tiene trabajar con herramientas de ayuda al diseño es que se pierde contacto rápidamente con el problema, en la forma en que fue escrito por el diseñador. Esto se debe a varias causas. Los compiladores suelen proceder a través de la generación de numerosos archivos intermedios y además usan innumerables archivos auxiliares con información; de los archivos generados para la compilación del diseño, algunos de ellos son binarios, otros con formatos adecuados a los datos y tablas del programa y no son legibles con facilidad por un humano; además si existen archivos de texto, los rótulos e identificadores son elegidos por un autómatas, y son de difícil asociación para el programador. Es decir, prácticamente se pierde control del proceso de síntesis, sólo puede verse lo que la aplicación está diseñada para mostrar: resúmenes del proceso de síntesis, esquemas, etc.

Veremos el proceso de síntesis a través de algunos ejemplos.

Ejemplo. Síntesis de descripciones estructurales implícitas.

```
// Módulo combinacional que genera:  $f = ab + ac + ad$ 
module simple(output x, input a, b, c, d);
    assign x = a & b | a & c | a & d;
endmodule
```

La síntesis rtl genera un módulo en base a compuertas. Los nombres de las compuertas y los nodos internos han sido elegidos por un ser humano, y por esto el programa resulta fácil de leer.

```
module simpleRtl(output x, input a, b, c, d);
    wire n1, n2, n3;
    and a1 (n1, a, b);
    and a2 (n2, a, c);
    and a3 (n3, a, d);
    or o1 (x, n1, n2, n3);
endmodule
```

Sin embargo es posible que diferentes programas de síntesis generen implementaciones diferentes, por ejemplo usando sólo compuertas de fan-in 2, se obtendría:

```
module simpleRtlFain2(output x, input a, b, c, d);
  wire n1, n2, n3, n4;
  and a1 (n1, a, b);
  and a2 (n2, n1, n2);
  or o1 (n3, n1, n2);
  and a3 (n4, a, d);
  or o1 (x, n3, n4);
endmodule
```

Para ilustrar la rapidez con que la información detallada, de lo que se está realizando, se vuelve cada vez más incomprensible, se muestra la misma información del programa anterior, representada en un formato netlist xnf de xilinx (formato ya obsoleto), y generado por el compilador icarus, mediante el comando: iverilog -o simple.xnf -t xnf simple.v

```
LCANET,6
PROG,verilog,$Name: v0_8 $,"Icarus Verilog"
PART,
SIG, a, PIN=a
SIG, b, PIN=b
SIG, c, PIN=c
SIG, d, PIN=d
SIG, x, PIN=x
SYM, _s1, AND, LIBVER=2.0.0
  PIN, O, O, _s0
  PIN, I0, I, a
  PIN, I1, I, b
END
SYM, _s3, AND, LIBVER=2.0.0
  PIN, O, O, _s2
  PIN, I0, I, a
  PIN, I1, I, c
END
SYM, _s5, OR, LIBVER=2.0.0
  PIN, O, O, _s4
  PIN, I0, I, _s0
  PIN, I1, I, _s2
END
SYM, _s7, AND, LIBVER=2.0.0
  PIN, O, O, _s6
  PIN, I0, I, a
  PIN, I1, I, d
END
SYM, _s9, OR, LIBVER=2.0.0
  PIN, O, O, x
  PIN, I0, I, _s4
```

```

    PIN, I1, I, _s6
END
EOF

```

También se producen formatos netlist estandarizados, pero que son de difícil lectura para un ser humano.

Si el dispositivo físico, en que será depositado el diseño, tiene celdas básicas or y and de cuatro entradas, el módulo mapeado tecnológicamente y optimizado, resulta:

```

module simpleOptMapeado(output x, input a, b, c, d);
    wire n1;
    or4 o1 (n1, b, c, d, 1'b0);
    and4 a1 (x, a, n1, 1'b1, 1'b1);
endmodule

```

Esto debido a que $f = ab + ac + ad$, puede (y debería) ser minimizada a: $f = a(c + b + d)$.

Si el dispositivo sólo tiene compuertas inversoras y nor de 2 y 3 entradas, se obtendría lo siguiente, asumiendo que el sintetizador optimiza las ecuaciones booleanas:

```

module simpleOptMapeadoOr(output x, input a, b, c, d);
    wire n1, n2;
    inv i1 (n1, a);
    nor3 no1 (n2, b, c, d);
    nor2 no2 (x, n1, n2);
endmodule

```

Si el sistema es una fpga, en base a luts de 4 entradas, basta una para implementar la tabla de verdad de la función de cuatro variables.

Ejemplo. Síntesis de operaciones aritméticas en base a sumadores y muxs.

Se tienen dos buses de entrada de 8 bits cada uno, y se desea generar una salida cuando la suma de los operandos de entrada sea menor que 128 decimal.

El siguiente módulo describe con precisión una operación aritmética sencilla, mostrando la capacidad de descripción empleando lenguajes. Sin embargo el problema no es sencillo de sintetizar con métodos de papel y lápiz, ya que se tienen 16 entradas, lo cual implica una tabla de verdad con 65536 entradas, o mapas de Karnaugh de 16 variables.

Se ha empleado el macro `define para parametrizar el módulo. Su ventaja, en lugar de usar parameter, es que la definición puede ser empleada en varios módulos.

```

`define N 7
`define MAX 255
module aritmetica(x, a, b);
    input [N:0] a, b;

```

```

    output x;
    assign x= a+b < 128;
endmodule

```

El desarrollo un banco de prueba, para generar estímulos, básicamente consiste en generar todas las combinaciones posibles de las entradas; esto se logra con dos for anidados.

Puede notarse que el diseño del módulo test es muy similar a un programa escrito en C. Como el programa debe generar datos, posiblemente en la pantalla con fines de depuración y también en archivos para generar formas de ondas, gran parte del código es para lograr este objetivo.

Es en estos programas que se debe usar comandos que generen eventos, para que la simulación pueda ir avanzando.

```

module test(a, b, result);
output [^N:0] a, b;
input result;
reg [^N:0] a, b;
integer j, k;
integer errores, vectores;

```

```

initial
begin
    $dumpfile("mul.vcd");
    $dumpvars(0,a,b,result);
end

```

```

initial
begin
    //$monitor("a=%d b=%d result=%b", a, b, result, $time); //para depuración
    a=0; b=0;
    errores=0; vectores=0;
    for(j=0;j<`MAX;j=j+1)
    begin
        #10 a<=j;
        for(k=0;k<`MAX;k=k+1)
        begin
            #10 b<=k;
            vectores=vectores+1;
            if(result!=((a+b)<128) )
            begin
                errores=errores+1;
                $display("Error: a=%d b=%d result=%d Correcto=%d", a, b, result,(a+b)<128);
            end
        end
    end
end

```

```

$display("Se probaron %d vectores. Errores=%d", vectores, errores);

```

```

    #10 $finish;
end

endmodule

module BancoDePrueba;
wire [N:0] a, b;
wire x;
  aritmetica a1 (x, a, b);
  test      t1 (a, b, x); //las salidas de test alimentan al módulo aritmético.
endmodule

```

Debe notarse que los módulos para la prueba tienen bastantes más líneas que el módulo que se desea probar.

Con la ayuda de este test, se visualiza que hay problemas con el diseño, ya que se producen demasiados errores. Si se analizan se verá que la suma se está realizando en 8 bits, de este modo si los operandos son mayores que 128, se tendrá rebalse del sumador. Este puede corregirse si se especifica un sumador que entregue el resultado en 9 bits, de esta manera la suma máxima: $255+255=510$ no produce rebalse; es decir el máximo valor de la suma se puede representar en 9 bits, ya que $510 = 1'b111111110$. Esto asumiendo números sin signo.

Nos aseguramos que la suma se efectúe en 9 bits, definiendo el bus temp, esto implica cambios en el módulo:

```

module aritmetica(x, a,b);
  input [N:0] a, b;
  output x;
  wire [9:0] temp;
  assign temp=a+b; //el resultado de la suma se calcula con 9 bits
  assign x= temp < 128;
endmodule

```

La herramienta de síntesis podría haber reconocido lo anterior, y emplear dos módulos de sus bibliotecas: un sumador de 9 bits, y un circuito comparador slt (por set less than) que genere la señal de salida.

```

module aritmeticaSint(x, a,b);
  input [N:0] a, b;
  output x;
  wire [9:0] temp;
  sumador s1 (temp, a, b); //operandos de 8 bits, suma con reserva de salida.
  slt  sl1 (x, temp, 9d'128); //operandos de 9 bits
endmodule

```

También este camino podría seguir un diseñador, permitiendo descomponer el diseño, en dos módulos más simples. Según muestra el esquemático de la Figura A5.44




Figura A5.44

Ahora deben desarrollarse dos nuevos módulos, los cuales pueden simularse por separado hasta verificar su funcionalidad. El sumador es posible que sea mejor sintetizado por la herramienta CAD, ya que los sumadores son un recurso fundamental en sistemas digitales, y se ha trabajado suficiente en su desarrollo para optimizar la velocidad de éstos.

El comparador también debe estar en las bibliotecas del sintetizador; pero podría ser bastante más económico en espacio, el realizarlo como un módulo combinacional, ya que en este caso sólo se tienen 9 entradas, en lugar de 18. Además el número $128 = 1'b01000000$, es una potencia de dos, y se podría diseñar el módulo combinacional, considerando que los dos primeros bits de temp deben ser ceros, para que éste sea menor que 128. Emplearemos esta idea, que difícilmente sería realizada por un autómata de síntesis digital, definiendo un módulo combinacional slt128 de sólo dos entradas, los dos bits más significativos del bus temp.

```

module aritmeticaSint2(x, a, b);
    input [N:0] a, b;
    output x;
    wire [9:0] temp;
    assign temp=a+b; //operandos de 8 bits, suma con reserva de salida en 9 bits.
    slt128 s1 (x, temp[8:7]);
endmodule

```

El esquemático con la arquitectura de aritmeticaSint2, se muestra en la Figura A5.45.




Figura A5.45

Asumiendo que el sintetizador es eficiente, se le podría delegar la responsabilidad de generar slt128, mediante una descripción conceptual de la operación: los dos bit de entrada deben ser ceros.

```

module slt128(output x, input [1:0] val);
  assign x= ( (val[1] | val[0]) ==0);
endmodule

```

Evaluando el costo de este módulo, puede no ser necesario seguir descendiendo en el nivel de detalle. Pero si se desea asegurar un diseño mínimo, puede escribirse a nivel de compuertas, según:

```

module slt128(output x, input [1:0] val);
  and(x, ~val[1], ~val[0]);
endmodule

```

Conviene conocer el diseño en bajo nivel de los subsistemas típicamente usados, de esta manera se conocerá su funcionalidad, sus alcances, la forma de generalizarlos, y su costo. Luego las descripciones de la arquitectura pueden realizarse basada en esos módulos.

Un curso básico en sistemas digitales debe profundizar en el **diseño de los bloques básicos**, y también capacitar en **descomponer un problema en términos de esos bloques**. Esto garantiza plantear arquitecturas con partes que serán sintetizadas eficientemente.

La Figura A5.46, muestra el proceso de síntesis como un nivel intermedio, entre las descripciones de arquitecturas mediante módulos sintetizables (top-down), y el desarrollo de módulo básicos implementados mediante compuertas y flip-flops (bottom-up). Los lenguajes de descripción de hardware, como Verilog, permiten las descripciones y simulaciones en los diferentes niveles, dando homogeneidad al proceso de diseño.




Figura A5.46

Referencias.

Tutoriales y conceptos sobre Verilog.

Nonblocking Assignments in Verilog Synthesis, Coding Styles That Kill! Clifford E. Cummings. SNUG 2000.

<http://www.see.ed.ac.uk/~gerard/Teach/Verilog/manual/Intro/gatetypes.html>

http://www.doulos.com/knowhow/verilog_designers_guide/

Cursos universitarios sobre Verilog.

El profesor David M. Koppelman, de Louisiana State University, Baton Rouge, tiene excelentes materiales para aprender Verilog, aplicado a sistemas digitales y estructuras de computadores en: <http://www.ece.lsu.edu/ee3755/ln.html>.

El profesor Weidong Kuang, The University of Texas, desarrolló un curso básico de sistemas digitales en: <http://www.engr.panam.edu/~kuangw/courses/ELEE4303/index.html>.

Software libre.

Herramienta de simulación y síntesis.

Icarus Verilog

<http://www.icarus.com/eda/verilog/index.html>

Visor de formas de ondas.

GTKWave visor de archivos VCD/EVCD. Anthony J Bybell .

<http://www.geocities.com/SiliconValley/Campus/3216/GTKWave/gtkwave-win32.html>

Desgraciadamente las referencias a la red son volátiles y no permanentes.

Índice general.

APÉNDICE 5	1
USO DE VERILOG	1
A5.1 INTRODUCCIÓN	1
A5.2 DESCRIPCIÓN ESTRUCTURAL	2
A5.2.1. Nivel compuertas.....	2
Ejemplo A5.1.....	4
Ejemplo A5.2.....	4
A5.2.2. Nivel transistores.....	5
Ejemplo A5.3.....	5
Ejemplo A5.4.....	6
A5.3 DESCRIPCIÓN DEL COMPORTAMIENTO (BEHAVIOR).	7
Ejemplo A5.5. Expresiones.	8
Ejemplo A5.6. Mux mediante Expresiones.	8
Operadores.	8
Ejemplo A5.7. Procesos.	9
Ejemplo A5.8. Mux mediante Procesos.	10
Ejemplo A5.9. Mux mediante lenguaje.	10
Sentencias.	11
A5.4 DISEÑOS JERÁRQUICOS.	12
A5.5. FUNDAMENTOS DE SIMULACIÓN.....	13
Ejemplo A5.10. Banco de prueba para semisumador binario.	14
Funcionamiento del simulador.	15
Ejemplo A5.11. Simulación nivel transistores.	16
A5.6. BUSES, REGISTROS, VECTORES, ARREGLOS. VALORES CONSTANTES.	17
Valores.....	18
Ejemplo A5.12. Mux con dos buses de entrada.	19
Ejemplo A5.13. Mux con varios buses de entrada.	20
Ejemplo A5.14. Mux con operador condicional.	20
Ejemplo A5.15. ALU con operandos vectores.	22
Ejemplo A5.16. Sumador completo con operandos vectores.....	22
Ejemplo A5.17. Estructuras repetitivas. Lazos for dentro de procesos.	22
Ejemplo A5.18. Descripción algorítmica, empleando operadores del lenguaje.....	23
Ejemplo A5.19. Decodificador, empleando operadores del lenguaje.....	24
Ejemplo A5.20. Buffer de tercer estado.....	25
Ejemplo A5.21. Relación entre simulación y síntesis.	26
A5.7 SISTEMAS SECUENCIALES.....	26
Latch transparente. (D gated latch).....	26
Flip-flop D. Disparado por canto de subida.	31
Flip-flop D. Disparado por canto de bajada.	32
Flip-flop T.....	32
Controles de reset asincrónicos y sincrónicos.....	33
Registro n bits.....	34
Carreras en simulación.	35
Registro de desplazamiento.	35
Contador ascendente con reset de lógica negativa.	36

<i>Contador con reset, carga paralela y habilitación para contar.....</i>	<i>37</i>
<i>Divisor de frecuencia por n.....</i>	<i>38</i>
<i>Transferencias entre registros de segmentación en pipeline.....</i>	<i>39</i>
<i>Registro de desplazamiento con realimentaciones lineales.</i>	<i>40</i>
<i>Contador de anillo</i>	<i>42</i>
<i>Contador BCD de dos cifras.</i>	<i>43</i>
A5.8. MÁQUINAS SECUENCIALES DE ESTADOS FINITOS.	44
<i>Diagramas de Moore.</i>	<i>44</i>
<i>Diagramas de Mealy.</i>	<i>46</i>
<i>Ejemplo.</i>	<i>47</i>
A5.9. SISTEMAS ASINCRÓNICOS.	48
<i>Latch de nand.....</i>	<i>48</i>
<i>Descripción estructural flip-flop D, en base a nand.</i>	<i>49</i>
<i>Flip-flop RS.</i>	<i>50</i>
A5.10. SÍNTESIS.	50
<i>Ejemplo. Síntesis de descripciones estructurales implícitas.</i>	<i>52</i>
<i>Ejemplo. Síntesis de operaciones aritméticas en base a sumadores y muxs.</i>	<i>54</i>
REFERENCIAS.	60
<i>Tutoriales y conceptos sobre Verilog.....</i>	<i>60</i>
<i>Cursos universitarios sobre Verilog.....</i>	<i>60</i>
<i>Software libre.</i>	<i>60</i>
ÍNDICE GENERAL.	61
ÍNDICE DE FIGURAS.	63

Índice de Figuras.

Figura A5.1.....	2
Figura A5.2.....	3
Figura A5.3.....	4
Figura A5.4.....	5
Figura A5.5.....	6
Figura A5.6.....	7
Figura A5.7.....	12
Figura A5.8.....	13
Figura A5.9.....	14
Figura A5.10.....	18
Figura A5.11.....	18
Figura A5.12. mux4to1.	21
Figura A5.13.....	24
Figura A5.14.....	25
Figura A5.15.....	26
Figura A5.16.....	27
Figura A5.17.....	27
Figura A5.18.....	28
Figura A5.19.....	28
Figura A5.20.....	28
Figura A5.21.....	30
Figura A5.22.....	31
Figura A5.23.....	31
Figura A5.24.....	32
Figura A5.25.....	32
Figura A5.26.....	32
Figura A5.27.....	33
Figura A5.28.....	33
Figura A5.29.....	34
Figura A5.30.....	36
Figura A5.31.....	37
Figura A5.32.....	37
Figura A5.33.....	39
Figura A5.34.....	40
Figura A5.35.....	40
Figura A5.36.....	42
Figura A5.37.....	42
Figura A5.38.....	44
Figura A5.39.....	44
Figura A5.39.....	46
Figura A5.40.....	47
Figura A5.41.....	48
Figura A5.42.....	50
Figura A5.43.....	50

Figura A5.44.....	57
Figura A5.45.....	58
Figura A5.46.....	59